Problem 2

The regression results are presented in Table 1. Column 1 gives the estimates from the standard OLS regression, whereas column 2 gives the results from the OLS estimation using the white estimator of the covariance matrix. The FGLS estimates are listed in column 3, where Harvey's formulation[1] is used.

	OLS	OLS (White)	FGLS
	b/t	b/t	b/t
x1	1.139	1.139*	0.744
	(1.16)	(2.08)	(1.25)
x2	0.374	0.374	0.981**
	(0.85)	(0.34)	(2.78)
cons	0.192	0.192	0.200
	(0.21)	(0.26)	(0.28)
R^2-adj	-0.003	-0.003	0.113
N	50.000	50.000	50.000

TABLE 1. Results

Note: * p<0.05, ** p<0.01, *** p<0.001

The covariance matrix VCV gives the conventional estimator, whereas the covariance matrix VCV^* is the White estimator for the covariance matrix, which is used to in Table 1 column 2.

$$VCV\left(\hat{\beta}\right) = \begin{pmatrix} 0.9662 & 0.0511 & -0.1155 \\ 0.0511 & 2.1937 & -0.0472 \\ -0.1155 & -0.0472 & 0.8368 \end{pmatrix}$$

$$VCV^*\left(\hat{\beta}\right) = \begin{pmatrix} 0.3007 & -0.0972 & 0.0819 \\ -0.0972 & 1.2163 & 0.4239 \\ 0.0819 & 0.4239 & 0.5590 \end{pmatrix}$$

Two tests for heteroskedasticity were carried out - White's general test and the Breusch Pagan test (Koenker variation). Both tests reject H_0 of homoskedasticity.

	White	BP
LM	39.0961	14.1844
p-val	0.0000	0.0008

TABLE 2. Tests for Heteroskedasticity

Note: $LM = nR^2$

[1] $Var[\epsilon_i|x_{i1}, x_{i2}] = \sigma^2 \exp(\gamma_1 x_{i1} + \gamma_2 x_{i2})$

Looking at the regression results of these two tests (Table 3), indicates a relationship between the squared residuals and the explanatory variable x_2. The estimation when applying White's general tests suggest that this relation is a quadratic one. This suggestion is supported by Figure 1, where the absolute values of the residuals are plotted against x_1 and x_2.

	WHITE	BP
	b/t	b/t
x1	6.032	4.395
	(0.64)	(0.27)
x1x1	0.603	
	(0.07)	
x1x2	8.374	
	(1.27)	
x2	3.839	31.720***
	(0.75)	(4.31)
x2x2	12.351***	
	(9.16)	
cons	-15.267	30.971*
	(-1.19)	(2.02)
R^2-adj	0.757	0.253
N	50.000	50.000

Table 3. Results

Note: * p<0.05, ** p<0.01, *** p<0.001



Figure 1. Absolute value of residuals versus explanatory variables

STATA code to Problem 2

```
cap log c
log using probset4, replace

* IMPORT TO STATA

insheet using data.txt, names t
compress
save probset4.dta, replace

* 1. OLS
* y = b0 + b1*x1 + b2*x2 + u (1)
reg y x1 x2
estat vce
est sto ols1
predict e, res
g e2 = e^2

* 2. OLS with white standard errors
reg y x1 x2, vce(robust)
estat vce
est sto ols2

* 1 & 2 by hand in mata

mata:
   mata clear
   y = st_data(.,("y"))
   X = st_data(.,("x1","x2"))
   ones = J(st_nobs(),1,1)
   X = (X,ones)

   bhat = invsym(X'*X)*X'*y

   // conventional variance estimate
```

```
   u_hat = (y - X*bhat)
   Vhat_con = u_hat'*u_hat/(rows(X) - cols(X))*cholinv(X'*X)

   // White-standard errrors
   Vhat_white = J(cols(X),cols(X),0)
   for (i=1; i<= rows(X); i++) {
      Vhat_white = Vhat_white +  (u_hat[i,1]*X[i,.]'*u_hat[i,1]*X[i,.])
   }
   Vhat_white = (rows(X)/(rows(X)-cols(X)))*invsym(X'*X)*Vhat_white*invsym(X'*X)

   bhat
   Vhat_con
   Vhat_white
end

est res ols1
predict absu, resid
replace absu = abs(absu)
qui twoway (scatter absu x1) (lowess absu x1, bw(0.4) lw(thick)), name(g1)
qui twoway (scatter absu x2) (lowess absu x2, bw(0.4) lw(thick)), name(g2)
graph combine g1 g2
gr export graph.png, replace


* 3. WHITE's general test
est res ols1

* implemented:
estat imtest, p w

* by hand:
g x1x1 = x1^2
g x2x2 = x2^2
g x1x2 = x1*x2
reg e2 x1* x2*
est sto e2
sca L = 5
sca LM = _N*e(r2)
sca p_val = chi2(LM,L)
di "White's general test by hand" _newline "LM = " LM _newline "p-value = " p_val

* 4. BREUSCH-PAGAN
est res ols1

* implemented: => type ssc install ivhettest
ivhettest, nr2

* by hand:
reg e2 x1 x2
est sto BP
sca L = 2
sca LM = _N*e(r2)
sca p_val = chi2(LM,L)
```

```
di "Breusch Pagan test (Koenker variation)" _newline "LM = " LM _newline "p-value = " p_val

* 5. FGLS
* 1. step: estimate (1) by OLS
* 2. step: regress squared residuals (sigma^2) on x1 x2 x1x2 x1x1 x2x2
* 3. reestimate (1) using 1/sigma as weights

gen loge2 = log(e2)
reg loge2 x1 x2, noc
predict sigma2, xb

replace sigma2 = exp(sigma2)

* implemented: regress with option weight
reg y x1 x2 [aweight=1/sigma2]
est sto fglsi

* by hand:
preserve
gen double ones = 1/sqrt(sigma2)
replace y  = y/sqrt(sigma2)
replace x1 = x1/sqrt(sigma2)
replace x2 = x2/sqrt(sigma2)
reg y x1 x2 ones, noc
est sto fglsh
restore

estout ols1 ols2 fglsi fglsh, cells(b(star fmt(3)) t(par fmt(2))) legend label stats(r2_a N)
estout ols1 ols2 fglsi, style(tex) cells(b(star fmt(3)) t(par fmt(2))) legend label stats(r2_a N)

estout e2 BP, style(tex) cells(b(star fmt(3)) t(par fmt(2))) legend label stats(r2_a N)

log c
exit
```